
[bookmark: _Toc238359817]This is an OGC® Candidate Standard for encoding binary representations of space-time varying georeferenced data.

Suggested additions, changes, and comments on this draft report are welcome and encouraged. Such suggestions may be submitted by email message or by making suggested changes in an edited copy of this document.

NetCDF (network Common Data Form) is a data model for array-oriented scientific data. There is a freely distributed collection of access libraries implementing support for that data model, and a machine-independent format. Together, the interfaces, libraries, and format support the creation, access, and sharing of scientific data.
[bookmark: _Toc250713807][bookmark: _Toc250904803][bookmark: _Toc250904993]Background information regarding the overall landscape of netCDF standards is presented in the CF-netCDF Primer, OGC 10-091r1, “CF-netCDF: Core and Extensions.” This specification is an extension to the core specification for the netCDF Classic data model in OGC 10-091r1, “NetCDF Core.”  

[bookmark: _Toc443461091][bookmark: _Toc443470360]OGC Binary Encoding Extension Standard: 
netCDF Classic and 64-bit Offset Format
[bookmark: _Toc238359823][bookmark: _Toc246317763][bookmark: _Toc250713809][bookmark: _Toc250904805][bookmark: _Toc250904995][bookmark: _Toc270671515]Scope
This document specifies the netCDF classic and 64-bit offset file binary encoding formats.  This standard specifies a set of requirements that every netCDF classic or 64-bit offset binary encoding must fulfil.   

[bookmark: _Toc250713817][bookmark: _Toc250904813][bookmark: _Toc250905003][bookmark: _Toc250905294][bookmark: _Toc443461092][bookmark: _Toc443470361][bookmark: _Toc238359824][bookmark: _Toc246317764][bookmark: _Toc250713818][bookmark: _Toc250904814][bookmark: _Toc250905004][bookmark: _Toc270671516]Conformance
Standardization targets are netCDF classic and 64-bit offset binary dataset encodings. 
This document establishes two requirements classes of:
· netCDF classic with URI http://www.opengis.net/spec/netcdf_binary-encoding_netcdf-classic-and-64-bit-offset/1.0/conf /netcdf-classic
· netCDF 64-bit offset with URI http://www.opengis.net/spec/netcdf_binary-encoding_netcdf-classic-and-64-bit-offset /1.0/conf /netcdf-64-bit-offset
· Requirements and conformance test URIs defined in this document are relative to http://www.opengis.net/spec/netcdf_binary-encoding_netcdf-classic-and-64-bit-offset /1.0.
Annex A (normative) specifies how data are encoded in netCDF classic and 64-bit offset binary formats.    In addition, these encodings must satisfy the tests listed the abstract test suite for the netCDF core in Annex A of OGC 10-090r1, “NetCDF Core.”  
[bookmark: _Toc443461093][bookmark: _Toc443470362][bookmark: _Toc238359825][bookmark: _Toc246317765][bookmark: _Toc250713826][bookmark: _Toc250904824][bookmark: _Toc250905014][bookmark: _Toc270671517]Normative references
[bookmark: _Toc443461094][bookmark: _Toc443470363][bookmark: _Toc238359826]The NetCDF Classic and 64-bit Binary Encoding Extension is contained within this document.  The specification is identified by OGC URI http://www.opengis.net/spec/netcdf_binary-encoding_netcdf-classic-and-64-bit-offset/1.0.
The document has OGC URL http://www.opengis.net/doc/netcdf_binary-encoding_netcdf-classic-and-64-bit-offset/1.0.
The following normative document contains provisions that, through reference in this text, constitute provisions of this specification. For undated references, the latest edition of the referenced document (including any amendments) applies.
NetCDF Core Specification.  OGC Document 10-091r1. http://www.opengis.net/doc/AS/netCDF
For this specification, there is one external normative document contain provisions that are quoted verbatim in this text and hence constitute provisions of this specification. 
NASA ESDS-RFC-011v2.00 R. Rew, E. Hartnett, D. Heimbigner, E. Davis, J. Caron: NetCDF Classic and 64-bit Offset File Formats
http://www.esdswg.org/spg/rfc/esds-rfc-011/ESDS-RFC-011v2.00.pdf
[bookmark: _Toc246317767][bookmark: _Toc250713829][bookmark: _Toc250904827][bookmark: _Toc250905017][bookmark: _Toc270671518]Terms and definitions
[bookmark: _Toc270671519]Definitions
For purposes of this document, the definitions in OGC 10-090r1, NetCDF Core, apply.
[bookmark: _Toc246317768][bookmark: _Toc250713830][bookmark: _Toc250904828][bookmark: _Toc250905018][bookmark: _Toc270671520]Acronyms (and abbreviated terms)
Some frequently used abbreviated terms:
API	Application Program Interface
BNF	Backus-Naur Form
CF	Climate and Forecast Conventions
ESDSWG	NASA Earth Standards Data Systems Working Groups
ES	Earth Sciences
GIS	Geographic Information System
HDF5	Hierarchical Data Format version 5
NcML	NetCDF Markup Language
NcML-GML	NetCDF Markup Language – Geography Markup Language
NetCDF	Network Common Data Form
NetCDF-4	NetCDF Release 4
ISO	International Organization for Standardization
OGC	Open Geospatial Consortium
SPG	NASA Standards Process Group
UML	Unified Modeling Language
XML	eXtended Markup Language
WFS	Web Feature Service
WCS	Web Coverage Service
1-D	One Dimensional
2-D	Two Dimensional
[bookmark: _Toc493581631][bookmark: _Toc238359827][bookmark: _Toc246317769][bookmark: _Toc250713831][bookmark: _Toc250904829][bookmark: _Toc250905019][bookmark: _Toc270671521][bookmark: _Toc443461095][bookmark: _Toc443470364]Document Conventions
[bookmark: _Toc482665220][bookmark: _Toc485456415][bookmark: _Toc493581633][bookmark: _Toc238359829][bookmark: _Toc246317770][bookmark: _Toc250713832][bookmark: _Toc250904830][bookmark: _Toc250905020][bookmark: _Toc270671522][bookmark: _Toc443461096][bookmark: _Toc443470365]UML Notation
The diagrams that appear in this standard are presented using the Unified Modeling Language (UML) static structure diagram.
[bookmark: _Toc270671523]BNF Notation	Comment by Ben Domenico: Note that MS Word often decides to reformat text that is supposed to be left “as is.”  This causes problems with the BNF sections, so they will have to be carefully checked at the end of the editing process to make sure that gratuitous word spacing and wrapping has not occurred.
To present the format more formally, we use a BNF grammar notation. In this notation:
· Non-terminals (entities defined by grammar rules) are in lower case.
· Terminals (atomic entities in terms of which the format specification is written) are in upper case, and are specified literally as US-ASCII characters within single-quote characters or are described with text between angle brackets (‘<’ and ‘>’).
· Optional entities are enclosed between braces (‘[’ and ‘]’).
· A sequence of zero or more occurrences of an entity is denoted by ‘[entity ...]’.
· A vertical line character (‘|’) separates alternatives. Alternation has lower precedence than concatenation.
· Comments follow ‘//’ characters.
· A single byte that is not a printable character is denoted using a hexadecimal number with the notation ‘\xDD’, where each D is a hexadecimal digit.
· A literal single-quote character is denoted by ‘\'’, and a literal back-slash character is denoted by ‘\\’.

Following the grammar, a few additional notes are included to specify format characteristics that are impractical to capture in a BNF grammar, and to note some special cases for implementers. Comments in the grammar point to the notes and special cases, and help to clarify the intent of elements of the format.
[bookmark: _Toc270671524]Namespace prefix conventions
The following namespaces are used in this document.  The prefix abbreviations used constituted conventions used here, but are not normative.  The namespaces to which the prefixes refer are normative, however.
Table 1 -- Namespace Mappings
	Prefix
	Namespace URI
	Description

	netCDF
	http://www.opengis.net/netcdf/
	netCDF 



[bookmark: _Toc246317771][bookmark: _Toc250713833][bookmark: _Toc250904831][bookmark: _Toc250905021][bookmark: _Toc270671525]netCDF Classic and 64-bit Offset File Formats Extension Standard
[bookmark: _Toc246317772][bookmark: _Toc250713834][bookmark: _Toc250904832][bookmark: _Toc250905022]This document formally specifies two format variants, the classic format and the 64-bit offset format for netCDF dataThe NetCDF Classic Data Model which is specified in OGC 10-NCD, “NetCDF Core.”  
Following are the requirements for the netCDF classic and 64-bit offset file format.  Understanding the format at this level can make clear which netCDF operations are expensive, for example adding a new variable to an existing file.

This standard defines two levels of conformance, as shown in the following diagram.
[image: Core netCDF-new.jpg]

Figure 1 Conformance classes and modules diagram
Referring to Figure 1, the following conformance classes and modules are defined, as detailed in the following paragraphs. Related conformance test cases are defined in Annex A.
	NetCDF Classic Data Model
(This is a conformance module for both the two conformance classes)
http://www.opengis.net/spec/netCDF/1.0/conf /core
	certifies the conformance to the abstract netCDF data model (array-oriented scientific data) requirement OGC 10-091r1, “NetCDF Core.”  

	NetCDF Classic Binary Format
(This is a conformance class)
http://www.opengis.net/spec/netcdf_binary-encoding_netcdf-classic-and-64-bit-offset/1.0/conf /netcdf-classic
	certifies the conformance to the classic binary format requirement 6.1.2

	NetCDF 64-bit Offset Format
(This is a conformance class)
http://www.opengis.net/spec/netcdf_binary-encoding_netcdf-classic-and-64-bit-offset/1.0/conf /netcdf-64-bit-offset
	certifies the conformance to the 64-bit variant of the classic binary format requirement 6.1.3

	Three Part File
(This is a conformance module for both the two conformance classes)
	certifies the requirement for a three part file requirement 6.1.2.1

	Header
(This is a conformance module for both the two conformance classes)
	certifies the conformance to the netCDF header requirement 6.1.2.2

	Fixed-size (non-record) Data
(This is a conformance module for both the two conformance classes)
	certifies the conformance to the netCDF fixed-size data requirement 6.1.2.3

	Record Data
(This is a conformance module for both the two conformance classes)
	certifies the conformance to the netCDF record data requirement 6.1.3.4


Table 2 Conformance class table

[bookmark: _Toc270671526]NetCDF Classic Abstract Data Model
[bookmark: Req01]Requirement 1 /req/classic-binary/data-model	The data shall conform to the netCDF classic abstract model as specified in the document OGC 10-NCD, “NetCDF Core.”  Related conformance test cases are defined in Annex A of OGC 10-NCD, “NetCDF Core.”   
[bookmark: _Toc270671527]NetCDF Classic Binary Dataset Format
The data shall conform to the netCDF classic file format as specified in the following sections. Related conformance test cases are defined in section A.1.
[bookmark: _Toc270671528]Three Part File
A classic or 64-bit offset file shall be stored in three parts: the header, the fixed-size (non-record) data, and the record data. Related conformance test cases are defined in section A.1.
[bookmark: Req02]Requirement 2 /req/classic-binary/netcdf-dataset-components/ 	A netCDF dataset shall have a header (header) section and a data (data) section.  
[bookmark: Req03]Requirement 3 /req/classic-binary/data-section-components/	The data section shall have a fixed –size, non-record (non-recs) and record (recs) section
[bookmark: Req04]Requirement 4 /req/classic-binary/header-part/	There shall be only one header part per file. 
[bookmark: Req05]Requirement 5 /req/classic-binary/fixed-size-data-part/	There shall be only one fixed-size data part per file. 
[bookmark: Req06]Requirement 6 /req/classic-binary/record-data-part/	There shall be only one record data part per file.
[bookmark: Req07]Requirement 7 /req/classic-binary/BNF-for-header-non-recore-record/	The header, non-record and record parts shall conform to the BNF grammar segment given below
	netcdf_file  = header data
   .
   .
   .
data         = non_recs recs



[bookmark: _Toc270671529]The Header
[bookmark: Req08]Requirement 8 /req/classic-binary/header-part-specifications/	The header shall specify: 
· whether classic or 64-bit offset encoding (magic) is used
· the length of the record dimension (numrecs)
· the list of dimenisons (dim_list)
· the list of global attributes (gatt_list)
· the list of variables (var_list) 

[bookmark: Req09]Requirement 9 /req/classic-binary/BNF-for-header/	The header shall conform to the BNF grammar segment given below.
	header       = magic numrecs dim_list gatt_list var_list
magic        = 'C' 'D' 'F' VERSION
VERSION      = \x01 |                   // classic format
numrecs      = NON_NEG | STREAMING      // length of record dimension
dim_list     = ABSENT | NC_DIMENSION nelems [dim ...]
gatt_list    = att_list                 // global attributes
att_list     = ABSENT | NC_ATTRIBUTE nelems [attr ...]
var_list     = ABSENT | NC_VARIABLE nelems [var ...]
ABSENT       = ZERO ZERO                // Means list is not present
ZERO         = \x00 \x00 \x00 \x00      // 32-bit zero
NC_DIMENSION = \x00 \x00 \x00 \x0A      // tag for list of dimensions
NC_VARIABLE  = \x00 \x00 \x00 \x0B      // tag for list of variables
NC_ATTRIBUTE = \x00 \x00 \x00 \x0C      // tag for list of attributes
nelems       = NON_NEG                  // number of elements in following sequence
dim          = name dim_length
name         = nelems namestring
		            // Names a dimension, variable, or attribute.
                   // Names should match the regular expression
                   //([a-zA-Z0-9_]|{MUTF8})([^\x00-\x1F/\x7F-\xFF]|{MUTF8})*
                   // For other constraints, see “Note on names”, below.
namestring   = ID1 [IDN ...] padding
ID1          = alphanumeric | '_'
IDN          = alphanumeric | special1 | special2
alphanumeric = lowercase | uppercase | numeric | MUTF8
lowercase    = 'a'|'b'|'c'|'d'|'e'|'f'|'g'|'h'|'i'|'j'|'k'|'l'|'m'|
               'n'|'o'|'p'|'q'|'r'|'s'|'t'|'u'|'v'|'w'|'x'|'y'|'z'
uppercase    = 'A'|'B'|'C'|'D'|'E'|'F'|'G'|'H'|'I'|'J'|'K'|'L'|'M'|
               'N'|'O'|'P'|'Q'|'R'|'S'|'T'|'U'|'V'|'W'|'X'|'Y'|'Z'
numeric      = '0'|'1'|'2'|'3'|'4'|'5'|'6'|'7'|'8'|'9'
                    // special1 chars have traditionally been
                    // permitted in netCDF names.
special1     = '_'|'.'|'@'|'+'|'-'
                    // special2 chars are recently permitted in
                    // names (and require escaping in CDL).
                    // Note: '/' is not permitted.
special2     = ' ' | '!' | '"' | '#' | '$' | '%' | '&' | '\'' |
               '(' | ')' | '*' | ',' | ':' | ';' | '<' | '=' |
               '>' | '?' | '[' | '\\' | ']' | '^' | '`' | '{' |
               '|' | '}' | '~'
MUTF8        = <multibyte UTF-8 encoded, NFC-normalized Unicode character>
dim_length   = NON_NEG       // If zero, this is the record dimension.
                             // There can be at most one record dimension.
attr         = name nc_type nelems [values ...]
nc_type      = NC_BYTE | NC_CHAR | NC_SHORT | NC_INT | NC_FLOAT | NC_DOUBLE
var          = name nelems [dimid ...] vatt_list nc_type vsize begin
                           // nelems is the dimensionality (rank) of the
                           // variable: 0 for scalar, 1 for vector, 2
                           // for matrix, ...
dimid        = NON_NEG     // Dimension ID (index into dim_list) for
                           // variable shape. We say this is a "record
                           // variable" if and only if the first
                           // dimension is the record dimension.
vatt_list    = att_list    // Variable-specific attributes
vsize        = NON_NEG     // Variable size. If not a record variable,
                             // the amount of space in bytes allocated to
                             // the variable's data. If a record variable,
                             // the amount of space per record. See “Note on
                             // vsize” below.
begin        = OFFSET        // Variable start location. The offset in
                             // bytes (seek index) in the file of the
                             // beginning of data for this variable.




[bookmark: _Toc270671530]The Fixed-size (Non-record) Data 
[bookmark: Req10]Requirement 10 /req/classic-binary/contiguous-for-fixed-size-data 	The data for all non-record variables shall be stored contiguously for each variable, in the same order the variables occur in the header.
[bookmark: Req11]Requirement 11 /req/classic-binary/block-values-for-fixed-size-data	All data for a non-record variable shall be stored as a block of values of the same type as the variable, in row-major order (last dimension varying fastest).
[bookmark: Req12]Requirement 12 /req/classic-binary/data-values-for non-record-variable   	The  fixed-size data shall contain data values for variables that don't have an unlimited dimension, i.e., for each non-record variable.  
[bookmark: Req13]Requirement 13 /req/classic-binary/data-row-major   	The data for each variable shall be stored contiguously, in row-major order for multi-dimensional variables. 
[bookmark: Req14]Requirement 14 /req/classic-binary/BNF-for-fixed-size-data  	The fixed-size (non-record) data shall conform to the BNF grammar segment given below.
	non_recs     = [vardata ...] // The data for all non-record variables,
                             // stored contiguously for each variable, in
                             // the same order the variables occur in the
                             // header.
vardata      = [values ...]  // All data for a non-record variable, as a
                             // block of values of the same type as the


                             // variable, in row-major order (last
                             // dimension varying fastest).



Related conformance test cases are defined in section A.1.
[bookmark: _Toc270671531] The Record Data
[bookmark: Req15]Requirement 15 /req/classic-binary/one-unlimted-dimension   	There shall be at most one unlimited dimension, the record dimension.
[bookmark: Req16]Requirement 16 /req/classic-binary/data-values-for-unlimted-dimension   	The  record data shall contain data values for variables that have an unlimited dimension.
[bookmark: Req17]Requirement 17 /req/classic-binary/current-size-in-header   	The current size of the record dimension shall be stored in the header which specifies how many records the file contains. 
[bookmark: Req18]Requirement 18 /req/classic-binary/all-data-for-record-part   	Each record in the record data part shall contain all the data for that record for each record variable. 
[bookmark: Req19]Requirement 19 /req/classic-binary/data-for-each-record   	Each record's worth of data for each record variable shall be stored contiguously, in row major order for multidimensional variables. 
[bookmark: Req20]Requirement 20 /req/classic-binary/record-size   	All records shall be the same size, because they each contain all the data for a particular record for each record variable. 
[bookmark: Req21]Requirement 21 /req/classic-binary/BNF-for-record-data   	The record data section shall conform to the BNF grammar segment given below.
	recs         = [record ...]  // The data for all record variables are
                             // stored interleaved at the end of the
                             // file.
record       = [varslab ...] // Each record consists of the n-th slab
                             // from each record variable, for example
                             // x[n,...], y[n,...], z[n,...] where the
                             // first index is the record number, which
                             // is the unlimited dimension index.
varslab      = [values ...]  // One record of data for a variable, a
                             // block of values all of the same type as
                             // the variable in row-major order (last
                             // index varying fastest).




Related conformance test cases are defined in section A.1.
[bookmark: _Toc270671532]BNF Definitions
[bookmark: Req22]Requirement 22 /req/classic-binary/BNF-definitions  	The BNF segments in the previous requirements shall conform to the BNF specifications in the segment below.
	values       = bytes | chars | shorts | ints | floats | doubles
string       = nelems [chars]
bytes        = [BYTE ...] padding
chars        = [CHAR ...] padding
shorts       = [SHORT ...] padding
ints         = [INT ...]
floats       = [FLOAT ...]
doubles      = [DOUBLE ...]
padding      = <0, 1, 2, or 3 bytes to next 4-byte boundary>
                             // Header padding uses null (\x00) bytes. In
                             // data, padding uses variable's fill value.
                             // See “Note on padding” below for a special
                             // case.
NON_NEG      = <non-negative INT>
STREAMING    = \xFF \xFF \xFF \xFF   // Indicates indeterminate record
                                     // count, allows streaming data
OFFSET       = <non-negative INT> |  // For classic format 
BYTE         = <8-bit byte>          // See “Note on byte data”, below.
CHAR         = <8-bit byte>          // See “Note on char data”, below.
SHORT        = <16-bit signed integer, Bigendian, two's complement>
INT          = <32-bit signed integer, Bigendian, two's complement>
INT64        = <64-bit signed integer, Bigendian, two's complement>
FLOAT        = <32-bit IEEE single-precision float, Bigendian>
DOUBLE       = <64-bit IEEE double-precision float, Bigendian>
                             // following type tags are 32-bit integers
NC_BYTE      = \x00 \x00 \x00 \x01       // 8-bit signed integers
NC_CHAR      = \x00 \x00 \x00 \x02       // text characters
NC_SHORT     = \x00 \x00 \x00 \x03       // 16-bit signed integers
NC_INT       = \x00 \x00 \x00 \x04       // 32-bit signed integers
NC_FLOAT     = \x00 \x00 \x00 \x05       // IEEE single precision floats
NC_DOUBLE    = \x00 \x00 \x00 \x06       // IEEE double precision floats
                           // Default fill values for each type, may be
                           // overridden by variable attribute named
                             // ‘_FillValue’, see “Note on fill values”, below
FILL_BYTE    = \x81                       // (signed char) -127
FILL_CHAR    = \x00                       // null byte
FILL_SHORT   = \x80 \x01                  // (short) -32767
FILL_INT     = \x80 \x00 \x00 \x01        // (int) -2147483647
FILL_FLOAT   = \x7C \xF0 \x00 \x00        // (float) 9.9692099683868690e+36
FILL_DOUBLE  = \x47 \x9E \x00 \x00 \x00 \x00 //(double)9.9692099683868690e+36





[bookmark: _Toc270671533]NetCDF 64-bit Offset Variant
The netCDF 64-bit offset format differs from the classic format only in the VERSION byte, \x02 instead of \x01, and the OFFSET entity, a 64-bit instead of a 32-bit offset from the beginning of the file. 
[bookmark: Req23]Requirement 23 /req/classic-binary/64-bit-offset    	A netCDF 64-bit Offset Variant dataset shall conform to all the requirements of the netCDF Classic encoding with the differences in the BNF grammar VERSION and OFFSET specifications given below.
	VERSION      =  \x02                   // 64-bit offset format

OFFSET       =  <non-negative INT64>  // for 64-bit offset format



This small format change permits much larger files, but there are still some practical size restrictions. Each fixed-size variable and the data for one record's worth of each record variable are still limited in size to a little less that 4 GiB. The rationale for this limitation is to permit aggregate access to all the data in a netCDF variable (or a record's worth of data) on 32-bit platforms. 
[bookmark: _Toc270671534]BNF Supplementary Notes
The following notes apply to the BNF segments in the specifications above.
Note on vsize: This number is the product of the dimension lengths (omitting the record dimension) and the number of bytes per value (determined from the type), increased to the next multiple of 4, for each variable. If a record variable, this is the amount of space per record. The netCDF “record size” is calculated as the sum of the vsize's of all the record variables. 
The vsize field is actually redundant, because its value may be computed from other information in the header. The 32-bit vsize field is not large enough to contain the size of variables that require more than 232 - 4 bytes, so 232 - 1 is used in the vsize field for such variables. 
Note on names: Earlier versions of the netCDF C-library reference implementation enforced a more restricted set of characters in creating new names, but permitted reading names containing arbitrary bytes. This RFC extends the permitted characters in names to include multi-byte UTF-8 encoded[7] Unicode[4] and additional printing characters from the US-ASCII alphabet. The first character of a name must be alphanumeric, a multi-byte UTF-8 character, or '_' (traditionally reserved for names with meaning to implementations, such as the “_FillValue” attribute). Subsequent characters may also include printing special characters, except for '/' which is not allowed in names. Names that have trailing space characters are also not permitted.
Implementations of the netCDF classic and 64-bit offset format must ensure that names are normalized according to Unicode NFC normalization rules [5] during encoding as UTF-8 for storing in the file header. This is necessary to ensure that gratuitous differences in the representation of Unicode names do not cause anomalies in comparing files and querying data objects by name.
Note on streaming data: The largest possible record count, 232-1, is reserved to indicate an indeterminate number of records. This means that the number of records in the file must be determined by other means, such as reading them or computing the current number of records from the file length and other information in the header. It also means that the numrecs􀀁field in the header will not be updated as records are added to the file. 
Note on padding: In the special case of only a single record variable of character, byte, or short type, no padding is used between data values. 
Note on byte data: It is possible to interpret byte data as either signed (-128 to 127) or unsigned (0 to 255). When reading byte data through an interface that converts it into another numeric type, the default interpretation is signed. There are various attribute conventions for specifying whether bytes represent signed or unsigned data, but no standard convention has been established. The variable attribute “_Unsigned” is reserved for this purpose in future implementations. 
Note on char data: Although the characters used in netCDF names must be encoded as UTF-8, character data may use other encodings. The variable attribute “_Encoding” is reserved for this purpose in future implementations.
Note on fill values: Because data variables may be created before their values are written, and because values need not be written sequentially in a netCDF file, default “fill values” are defined for each type, for initializing data values before they are explicitly written. This makes it possible to detect reading values that were never written. The variable attribute “_FillValue”, if present, overrides the default fill value for a variable. If _FillValue is defined then it should be scalar and of the same type as the variable. 

(normative)
Abstract Conformance Test Suite
A.1 [bookmark: _Toc250904881][bookmark: _Toc250905071][bookmark: _Toc270671537][bookmark: _Toc246317792][bookmark: _Toc250713883]Conformance Test Class: netCDF Classic Binary Format
A.1.1 [bookmark: _Toc270671538]Requirement 1
	Test ID
	Requirement 2 /req/classic-binary/netcdf-dataset-components

	Test purpose
	The data shall conform to the netCDF classic abstract model as specified in  OGC 10-NCD, “NetCDF Core.”  Related conformance test cases are defined in Annex A of OGC 10-NCD, “NetCDF Core.”

	Test method
	Verify that dataset satisfies the core conformance test cases defined in Annex A of OGC 10-NCD, “NetCDF Core.”


A.1.2 Requirement 2
	Test ID
	Requirement 2 /req/classic-binary/netcdf-dataset-components

	[bookmark: _Toc251683518][bookmark: _Toc251665974][bookmark: _Toc251669006][bookmark: _Toc251669290][bookmark: _Toc251669357]Test purpose
	A netCDF dataset shall have a header (header) section and a data (data) section.  

	Test method
	Open the dataset and verify that it has a header (header) section and a data (data) section.  


A.1.3 Requirement 3
	Test ID
	Requirement 3 /req/classic-binary/data-section-components

	Test purpose
	The data section shall have a fixed –size, non-record (non-recs) and record (recs) section

	Test method
	Open the dataset and verify that it has a fixed –size, non-record (non-recs) and record (recs) section.  


A.1.4 Requirement 4
	Test ID
	Requirement 4 /req/classic-binary/header-part

	Test purpose
	There shall be only one header part per file

	Test method
	Open the dataset and verify there is only one header part.  


A.1.5 [bookmark: _Toc270671542]Requirement 5
	Test ID
	Requirement 5 /req/classic-binary/fixed-size-data-part

	[bookmark: _Toc250904895][bookmark: _Toc250905085][bookmark: _Toc250905376][bookmark: _Toc250904896][bookmark: _Toc250905086][bookmark: _Toc250905377]Test purpose
	There shall be only one fixed-size data part per file

	Test method
	Open the dataset and verify that there is only one fixed-size data part


A.1.6 Requirement 6
	Test ID
	Requirement 6 /req/classic-binary/record-data-part

	Test purpose
	There shall be only one record data part per file

	Test method
	Open the dataset and verify there is only one record data part.  


A.1.7 Requirement 7	
	Test ID
	Requirement 7 /req/classic-binary/BNF-for-header-non-recore-record

	Test purpose
	The header, non-record and record parts shall conform to the  BNF grammar segment given in 6.1.2.

	Test method
	Open the dataset and verify that the header, non-record and record parts shall conform to the  BNF grammar segment given in 6.1.2.



A.1.8 Requirement 8	
	Test ID
	Requirement 8 /req/classic-binary/header-part-specifications

	Test purpose
	The header shall specify: 
· whether classic or 64-bit offset encoding (magic) is used
· the length of the record dimension (numrecs)
· the list of dimenisons (dim_list)
· the list of global attributes (gatt_list)
· the list of variables (var_list) 

	Test method
	Open the dataset and verify that the header specifies:   
· whether classic or 64-bit offset encoding (magic) is used
· the length of the record dimension (numrecs)
· the list of dimenisons (dim_list)
· the list of global attributes (gatt_list)
· the list of variables (var_list)




A.1.9  Requirement 9	
	Test ID
	Requirement 9 /req/classic-binary/BNF-for-header

	Test purpose
	The header shall conform to the BNF grammar segment given in 6.1.2.2.

	Test method
	Open the dataset and verify that the header conforms to the BNF grammar segment given in 6.1.2.2.



A.1.10 Requirement 10	
	Test ID
	Requirement 10 /req/classic-binary/contiguous-for-fixed-size-data

	Test purpose
	The data for all non-record variables shall be stored contiguously for each variable, in the same order the variables occur in the header

	Test method
	Open the dataset and verify that data for all non-record variables is stored contiguously for each variable, in the same order the variables occur in the header.



.
A.1.11 Requirement 11	
	Test ID
	Requirement 11 /req/classic-binary/block-values-for-fixed-size-data

	Test purpose
	All data for a non-record variable shall be stored as a block of values of  the same type as the variable, in row-major order (last dimension varying fastest).

	Test method
	Open the dataset and verify that all data for a non-record variable is stored as a block of values of the same type as the variable, in row-major order (last dimension varying fastest).



A.1.12 Requirement 12	
	Test ID
	Requirement 12 /req/classic-binary/data-values-for non-record-variable   

	Test purpose
	The fixed-size data shall contain data values for variables that don't have an unlimited dimension, i.e., for each non-record variable.  

	Test method
	Open the dataset and verify that the fixed-size data contain data values for variables that don't have an unlimited dimension, i.e., for each non-record variable.    



A.1.13 Requirement 13	
	Test ID
	Requirement 13 /req/classic-binary/data-row-major   

	Test purpose
	The data for each variable shall be stored contiguously, in row-major order for multi-dimensional variables. 

	Test method
	Open the dataset and verify that the data for each variable is stored contiguously, in row-major order for multi-dimensional variables.


A.1.14 Requirement 14	
	Test ID
	Requirement 14 /req/classic-binary/BNF-for-fixed-size-data  

	Test purpose
	The fixed-size (non-record) data shall conform to the BNF grammar segment given in 6.1.2.3.

	Test method
	Open the dataset and verify that the fixed-size (non-record) data conforms to the BNF grammar segment given in 6.1.2.3.



A.1.15 Requirement 15	
	Test ID
	Requirement 15 /req/classic-binary/one-unlimted-dimension   

	Test purpose
	There shall be at most one unlimited dimension, the record dimension.

	Test method
	Open the dataset and verify that there is at most one unlimited dimension and that it is the record dimension.



A.1.16 Requirement 16
	Test ID
	Requirement 16 /req/classic-binary/data-values-for-unlimted-dimension   

	Test purpose
	The record data shall contain data values for variables that have an unlimited dimension.

	Test method
	Open the dataset and verify that the record data contains data values for variables that have an unlimited dimension.



A.1.17 Requirement 17	
	Test ID
	Requirement 17 /req/classic-binary/current-size-in-header   

	Test purpose
	The current size of the record dimension shall be stored in the header which specifies how many records the file contains. 

	Test method
	Open the dataset and verify that the current size of the record dimension is stored in the header and specifies how many records the file contains.



A.1.18 Requirement 18	
	Test ID
	Requirement 18 /req/classic-binary/all-data-for-record-part   

	Test purpose
	Each record in the record data part shall contain all the data for that record for each record variable. 

	Test method
	Open the dataset and verify that each record in the record data part  contains all the data for that record for each record variable.



A.1.19 Requirement 19
	Test ID
	Requirement 19 /req/classic-binary/data-for-each-record   

	Test purpose
	Each record's worth of data for each record variable shall be stored contiguously, in row major order for multidimensional variables.

	Test method
	Open the dataset and verify that each record's worth of data for each record variable is stored contiguously, in row major order for multidimensional variables.




A.1.20 Requirement 20	
	Test ID
	Requirement 20 /req/classic-binary/record-size   

	Test purpose
	All records shall be the same size, because they each contain all the data for a particular record for each record variable. 

	Test method
	Open the dataset and verify that all records are the same size and that they each contain all the data for a particular record for each record variable.



A.1.21 Requirement 21	
	Test ID
	Requirement 21 /req/classic-binary/BNF-for-record-data   

	Test purpose
	The record data section shall conform to the BNF grammar segment given in 6.1.2.4.

	Test method
	Open the dataset and verify that record data section conforms to the BNF grammar segment given in 6.1.2.4.



A.1.22 Requirement 22	
	Test ID
	Requirement 22 /req/classic-binary/BNF-definitions  

	Test purpose
	The BNF segments in the previous requirements shall conform to the BNF specifications in 6.1.2.5.

	Test method
	Open the dataset and verify that BNF segments in the previous requirements conform to the BNF specifications in 6.1.2.5.



A.2 [bookmark: _Toc270671560]Test Class: netCDF 64-bit Offset Binary Format
A.2.1 Requirement 23	
	Test ID
	Requirement 23 /req/classic-binary/64-bit-offset    

	Test purpose
	A netCDF 64-bit Offset Variant dataset shall conform to all the requirements of the netCDF Classic encoding with the differences in the BNF grammar VERSION and OFFSET specifications given in 6.1.3.

	Test method
	Open the netCDF 64-bit Offset Variant dataset and verify that it conforms to all the requirements of the netCDF Classic encoding as well as the differences in the BNF grammar VERSION and OFFSET specifications given in 6.1.3.



Annex B [bookmark: _Toc251669300][bookmark: _Toc251669367][bookmark: _Toc250713892][bookmark: _Toc250904899][bookmark: _Toc250905089][bookmark: _Toc250905380][bookmark: _Toc270671562]  
(normative) 
Complete BNF Grammar
2. [bookmark: _Toc270671563][bookmark: _Toc250905381]Complete BNF Grammar for netCDF Classic and 64-bit Offset Binary Encoding
Note that this BNF grammar and the portions of it in the document body are verbatim quotes from 
 NASA ESDS-RFC-011v2.00 R. Rew, E. Hartnett, D. Heimbigner, E. Davis, J. Caron: NetCDF Classic and 64-bit Offset File Formats
http://www.esdswg.org/spg/rfc/esds-rfc-011/ESDS-RFC-011v2.00.pdf

To present the format more formally, we use a BNF grammar notation. In this notation:
· Non-terminals (entities defined by grammar rules) are in lower case.
· Terminals (atomic entities in terms of which the format specification is written) are in upper case, and are specified literally as US-ASCII characters within single-quote characters or are described with text between angle brackets (‘<’ and ‘>’).
· Optional entities are enclosed between braces (‘[’ and ‘]’).
· A sequence of zero or more occurrences of an entity is denoted by ‘[entity ...]’.
· A vertical line character (‘|’) separates alternatives. Alternation has lower precedence than concatenation.
· Comments follow ‘//’ characters.
· A single byte that is not a printable character is denoted using a hexadecimal number with the notation ‘\xDD’, where each D is a hexadecimal digit.
· A literal single-quote character is denoted by ‘\'’, and a literal back-slash character is denoted by ‘\\’.

Following the grammar, a few additional notes are included to specify format characteristics that are impractical to capture in a BNF grammar, and to note some special cases for implementers. Comments in the grammar point to the notes and special cases, and help to clarify the intent of elements of the format.

netcdf_file  = header data
header       = magic numrecs dim_list gatt_list var_list
magic        = 'C' 'D' 'F' VERSION
VERSION      = \x01 |                   // classic format
               \x02                     // 64-bit offset format
numrecs      = NON_NEG | STREAMING      // length of record dimension
dim_list     = ABSENT | NC_DIMENSION nelems [dim ...]
gatt_list    = att_list                 // global attributes
att_list     = ABSENT | NC_ATTRIBUTE nelems [attr ...]
var_list     = ABSENT | NC_VARIABLE nelems [var ...]
ABSENT       = ZERO ZERO                // Means list is not present
ZERO         = \x00 \x00 \x00 \x00      // 32-bit zero
NC_DIMENSION = \x00 \x00 \x00 \x0A      // tag for list of dimensions
NC_VARIABLE  = \x00 \x00 \x00 \x0B      // tag for list of variables
NC_ATTRIBUTE = \x00 \x00 \x00 \x0C      // tag for list of attributes
nelems       = NON_NEG                  // number of elements in following sequence
dim          = name dim_length
name         = nelems namestring
		            // Names a dimension, variable, or attribute.
                   // Names should match the regular expression
                   //([a-zA-Z0-9_]|{MUTF8})([^\x00-\x1F/\x7F-\xFF]|{MUTF8})*
                   // For other constraints, see “Note on names”, below.
namestring   = ID1 [IDN ...] padding
ID1          = alphanumeric | '_'
IDN          = alphanumeric | special1 | special2
alphanumeric = lowercase | uppercase | numeric | MUTF8
lowercase    = 'a'|'b'|'c'|'d'|'e'|'f'|'g'|'h'|'i'|'j'|'k'|'l'|'m'|
               'n'|'o'|'p'|'q'|'r'|'s'|'t'|'u'|'v'|'w'|'x'|'y'|'z'
uppercase    = 'A'|'B'|'C'|'D'|'E'|'F'|'G'|'H'|'I'|'J'|'K'|'L'|'M'|
               'N'|'O'|'P'|'Q'|'R'|'S'|'T'|'U'|'V'|'W'|'X'|'Y'|'Z'
numeric      = '0'|'1'|'2'|'3'|'4'|'5'|'6'|'7'|'8'|'9'
                    // special1 chars have traditionally been
                    // permitted in netCDF names.
special1     = '_'|'.'|'@'|'+'|'-'
                    // special2 chars are recently permitted in
                    // names (and require escaping in CDL).
                    // Note: '/' is not permitted.
special2     = ' ' | '!' | '"' | '#' | '$' | '%' | '&' | '\'' |
               '(' | ')' | '*' | ',' | ':' | ';' | '<' | '=' |
               '>' | '?' | '[' | '\\' | ']' | '^' | '`' | '{' |
               '|' | '}' | '~'
MUTF8        = <multibyte UTF-8 encoded, NFC-normalized Unicode character>
dim_length   = NON_NEG       // If zero, this is the record dimension.
                             // There can be at most one record dimension.
attr         = name nc_type nelems [values ...]
nc_type      = NC_BYTE | NC_CHAR | NC_SHORT | NC_INT | NC_FLOAT | NC_DOUBLE
var          = name nelems [dimid ...] vatt_list nc_type vsize begin
                           // nelems is the dimensionality (rank) of the
                           // variable: 0 for scalar, 1 for vector, 2
                           // for matrix, ...
dimid        = NON_NEG     // Dimension ID (index into dim_list) for
                           // variable shape. We say this is a "record
                           // variable" if and only if the first
                           // dimension is the record dimension.
vatt_list    = att_list    // Variable-specific attributes
vsize        = NON_NEG     // Variable size. If not a record variable,
                             // the amount of space in bytes allocated to
                             // the variable's data. If a record variable,
                             // the amount of space per record. See “Note on
                             // vsize” below.
begin        = OFFSET        // Variable start location. The offset in
                             // bytes (seek index) in the file of the
                             // beginning of data for this variable.
data         = non_recs recs
non_recs     = [vardata ...] // The data for all non-record variables,
                             // stored contiguously for each variable, in
                             // the same order the variables occur in the
                             // header.
vardata      = [values ...]  // All data for a non-record variable, as a
                             // block of values of the same type as the


                             // variable, in row-major order (last
                             // dimension varying fastest).
recs         = [record ...]  // The data for all record variables are
                             // stored interleaved at the end of the
                             // file.
record       = [varslab ...] // Each record consists of the n-th slab
                             // from each record variable, for example
                             // x[n,...], y[n,...], z[n,...] where the
                             // first index is the record number, which
                             // is the unlimited dimension index.
varslab      = [values ...]  // One record of data for a variable, a
                             // block of values all of the same type as
                             // the variable in row-major order (last
                             // index varying fastest).
values       = bytes | chars | shorts | ints | floats | doubles
string       = nelems [chars]
bytes        = [BYTE ...] padding
chars        = [CHAR ...] padding
shorts       = [SHORT ...] padding
ints         = [INT ...]
floats       = [FLOAT ...]
doubles      = [DOUBLE ...]
padding      = <0, 1, 2, or 3 bytes to next 4-byte boundary>
                             // Header padding uses null (\x00) bytes. In
                             // data, padding uses variable's fill value.
                             // See “Note on padding” below for a special
                             // case.
NON_NEG      = <non-negative INT>
STREAMING    = \xFF \xFF \xFF \xFF   // Indicates indeterminate record
                                     // count, allows streaming data
OFFSET       = <non-negative INT> |  // For classic format or
               <non-negative INT64>  // for 64-bit offset format
BYTE         = <8-bit byte>          // See “Note on byte data”, below.
CHAR         = <8-bit byte>          // See “Note on char data”, below.
SHORT        = <16-bit signed integer, Bigendian, two's complement>
INT          = <32-bit signed integer, Bigendian, two's complement>
INT64        = <64-bit signed integer, Bigendian, two's complement>
FLOAT        = <32-bit IEEE single-precision float, Bigendian>
DOUBLE       = <64-bit IEEE double-precision float, Bigendian>
                             // following type tags are 32-bit integers
NC_BYTE      = \x00 \x00 \x00 \x01       // 8-bit signed integers
NC_CHAR      = \x00 \x00 \x00 \x02       // text characters
NC_SHORT     = \x00 \x00 \x00 \x03       // 16-bit signed integers
NC_INT       = \x00 \x00 \x00 \x04       // 32-bit signed integers
NC_FLOAT     = \x00 \x00 \x00 \x05       // IEEE single precision floats
NC_DOUBLE    = \x00 \x00 \x00 \x06       // IEEE double precision floats
                           // Default fill values for each type, may be
                           // overridden by variable attribute named
                             // ‘_FillValue’, see “Note on fill values”, below
FILL_BYTE    = \x81                       // (signed char) -127
FILL_CHAR    = \x00                       // null byte
FILL_SHORT   = \x80 \x01                  // (short) -32767
FILL_INT     = \x80 \x00 \x00 \x01        // (int) -2147483647
FILL_FLOAT   = \x7C \xF0 \x00 \x00        // (float) 9.9692099683868690e+36
FILL_DOUBLE  = \x47 \x9E \x00 \x00 \x00 \x00 //(double)9.9692099683868690e+36

Note on vsize: This number is the product of the dimension lengths (omitting the record dimension) and the number of bytes per value (determined from the type), increased to the next multiple of 4, for each variable. If a record variable, this is the amount of space per record. The netCDF “record size” is calculated as the sum of the vsize's of all the record variables. 
The vsize field is actually redundant, because its value may be computed from other information in the header. The 32-bit vsize field is not large enough to contain the size of variables that require more than 232 - 4 bytes, so 232 - 1 is used in the vsize field for such variables. 
Note on names: Earlier versions of the netCDF C-library reference implementation enforced a more restricted set of characters in creating new names, but permitted reading names containing arbitrary bytes. This RFC extends the permitted characters in names to include multi-byte UTF-8 encoded[7] Unicode[4] and additional printing characters from the US-ASCII alphabet. The first character of a name must be alphanumeric, a multi-byte UTF-8 character, or '_' (traditionally reserved for names with meaning to implementations, such as the “_FillValue” attribute). Subsequent characters may also include printing special characters, except for '/' which is not allowed in names. Names that have trailing space characters are also not permitted.
Implementations of the netCDF classic and 64-bit offset format must ensure that names are normalized according to Unicode NFC normalization rules [5] during encoding as UTF-8 for storing in the file header. This is necessary to ensure that gratuitous differences in the representation of Unicode names do not cause anomalies in comparing files and querying data objects by name.
Note on streaming data: The largest possible record count, 232-1, is reserved to indicate an indeterminate number of records. This means that the number of records in the file must be determined by other means, such as reading them or computing the current number of records from the file length and other information in the header. It also means that the numrecs􀀁field in the header will not be updated as records are added to the file. 
Note on padding: In the special case of only a single record variable of character, byte, or short type, no padding is used between data values. 
Note on byte data: It is possible to interpret byte data as either signed (-128 to 127) or unsigned (0 to 255). When reading byte data through an interface that converts it into another numeric type, the default interpretation is signed. There are various attribute conventions for specifying whether bytes represent signed or unsigned data, but no standard convention has been established. The variable attribute “_Unsigned” is reserved for this purpose in future implementations. 
Note on char data: Although the characters used in netCDF names must be encoded as UTF-8, character data may use other encodings. The variable attribute “_Encoding” is reserved for this purpose in future implementations.
Note on fill values: Because data variables may be created before their values are written, and because values need not be written sequentially in a netCDF file, default “fill values” are defined for each type, for initializing data values before they are explicitly written. This makes it possible to detect reading values that were never written. The variable attribute “_FillValue”, if present, overrides the default fill value for a variable. If _FillValue is defined then it should be scalar and of the same type as the variable. 
Fill values are not required, however, because netCDF libraries have traditionally supported a “no fill” mode when writing, omitting the initialization of variable values with fill values. This makes the creation of large files faster, but also eliminates the possibility of detecting the inadvertent reading of values that were not written. 

	
	


1
	2
	



image1.jpeg
<<l part>>
Header

implements

o [ <<riepart>>
Fixed-size Data

1 [<<Fiepart>>
Record Data

Three part File.





